



TECNOLÓGICO DE ESTUDIOS SUPERIORES DE HUIXQUILUCAN

Sesión # 7 Cálculo Vectorial

Dr. Ildebrando Pérez Reyes

"Maple is a trademark of Waterloo Maple Inc."

"The computations in this paper were performed by using Maple(TM)."

Implementar cálculo vectorial en MAPLE es relativamente sencillo, ya que cuenta con una paquetería especial para realizar éste tipo de tareas. Para cuestiones académicas, se cuenta con una subpaquetería denominada **VectorCalculus** en **Student**

Veamos,

```
> with(Student[VectorCalculus]);  
[&x, `*`, `+`, `^`, `.` , < , > , <|> , About, ArcLength, BasisFormat, Binormal, ConvertVector,  
CrossProduct, Curl, Curvature, D, Del, DirectionalDiff, Divergence, DotProduct,  
FlowLine, Flux, GetCoordinates, GetPVDDescription, GetRootPoint, GetSpace, Gradient,  
Hessian, IsPositionVector, IsRootedVector, IsVectorField, Jacobian, Laplacian, LineInt,  
MapToBasis, Nabla, Norm, Normalize, PathInt, PlotPositionVector, PlotVector,  
PositionVector, PrincipalNormal, RadiusOfCurvature, RootedVector, ScalarPotential,  
SetCoordinates, SpaceCurve, SpaceCurveTutor, SurfaceInt, TNBFrame, Tangent,
```

(1)

TangentLine, TangentPlane, TangentVector, Torsion, Vector, VectorField, VectorFieldTutor, VectorPotential, VectorSpace, diff, evalVF, int, limit, series]

Es notable la cantidad de aplicaciones existentes.

Operaciones básicas en cálculo vectorial

Las operaciones en cálculo vectorial implican

- Suma
- Resta
- Multiplicación escalar
- Producto cruz

Escribir vectores es simple. Basta con escribir las componentes del vector separadas por comas y limitadas por $\langle \rangle$, por ejemplo

$$\begin{aligned} > \mathbf{v10} := \langle 2, 3, 4 \rangle; \\ & \mathbf{v10} := 2\mathbf{e}_x + 3\mathbf{e}_y + 4\mathbf{e}_z \end{aligned} \quad (1.1)$$

para coordenadas cartesianas, o si es el caso de que se esté usando algún otro sistema coordenado como el polar se hace lo siguiente

$$\begin{aligned} > \mathbf{v11} := \text{Vector}([2, 3], \text{coordinates}=\text{polar}[r, \text{theta}]); \\ & \mathbf{v11} := 2\mathbf{e}_r + 3\mathbf{e}_\theta \end{aligned} \quad (1.2)$$

De manera interesante es posible investigar acerca de los vectores $\mathbf{v10}$ y $\mathbf{v11}$ con el comando **About**

$$\begin{aligned} > \text{About}(\mathbf{v10}); \\ & \begin{array}{ll} \text{Type:} & \text{Free Vector} \\ \text{Components:} & [2, 3, 4] \\ \text{Coordinates:} & \text{cartesian} \end{array} \end{aligned} \quad (1.3)$$

$$\begin{aligned} > \text{About}(\mathbf{v11}); \\ & \begin{array}{ll} \text{Type:} & \text{Free Vector} \\ \text{Components:} & [2, 3] \\ \text{Coordinates:} & \text{polar}_{r, \theta} \end{array} \end{aligned} \quad (1.4)$$

Crear funciones vectoriales es posible con el comando VectorField como sigue

$$\begin{aligned} > \mathbf{F10} := \text{VectorField}(\langle x^2, 1+x*y \rangle, \text{cartesian}[x, y]); \\ & \mathbf{F10} := (x^2)\bar{\mathbf{e}}_x + (1 + xy)\bar{\mathbf{e}}_y \end{aligned} \quad (1.5)$$

Esto en el sistema cartesiano. Para coordenadas polares se tiene el siguiente ejemplo:

```
> F11:= VectorField(<r^2*cos(2*theta), r+sin(theta)>, polar[r, theta]);
```

$$F11 := (r^2 \cos(2\theta))\bar{e}_r + (r + \sin(\theta))\bar{e}_\theta \quad (1.6)$$

```
> About(F11);
```

$$\left[\begin{array}{ll} \text{Type:} & \text{Vector Field} \\ \text{Components:} & [r^2 \cos(2\theta), r + \sin(\theta)] \\ \text{Coordinates:} & \text{polar}_{r, \theta} \end{array} \right] \quad (1.7)$$


Considérese ahora graficar vectores en el plano. Comencemos primero por uno en 2D

```
> v12:= <5,4>;
```

$$v12 := 5e_x + 4e_y \quad (1.8)$$

Para graficar se usa el comando PlotVector

```
> PlotVector(v12, axes=boxed);
```




Ahora con uno en 3D

```
> v13:= <1,5,8>;
```

$$v13 := e_x + 5e_y + 8e_z$$

(1.9)

```
> PlotVector([v10,v13], axes=boxed, labels=[x,y,z]);
```



Pero obviamente se trata de vectores que siempre parten del inicio u origen. ¿Qué pasa cuando se trata de vectores que no parten del origen?

Este tipo de vectores MAPLE los conoce como rooted vectors. Para definirlos se hace como sigue:


```
> v14:= RootedVector(root=[2,9], [3,6]);
```

$$v14 := \begin{bmatrix} 3 \\ 6 \end{bmatrix}$$


(1.10)

veamos si funciona

```
> PlotVector(v14, axes=boxed);
```



Nótese que para escribir un vector de este tipo es necesario tener una noción clara de la suma y resta de vectores. Ya que debe tomarse en cuenta lo siguiente



Y de esta manera se pueden construir muchos otros vectores.

Pasemos ahora a la suma de vectores. Digamos que se desea sumar $\mathbf{v10}$ y $\mathbf{v12}$, entonces

$$\begin{aligned} & \text{[> v15:= <8,5>;} & v15 &:= 8e_x + 5e_y & (1.11) \end{aligned}$$

$$\begin{aligned} & \text{[> v15:= v12+v15;} & v15 &:= 13e_x + 9e_y & (1.12) \end{aligned}$$

$$\begin{aligned} & \text{[> v16:= v10+v13;} & v16 &:= 3e_x + 8e_y + 12e_z & (1.13) \end{aligned}$$

$$\begin{aligned} & \text{[> v17:= v10-v13;} & v17 &:= e_x - 2e_y - 4e_z & (1.14) \end{aligned}$$

Se puede realizar la multiplicación de un escalar por un vector,

$$\begin{aligned} & \text{[> v17a:= 5*v17;} & v17a &:= 5e_x - 10e_y - 20e_z & (1.15) \end{aligned}$$

$$\begin{aligned} & \text{[> v15a:= 4*v15;} & v15a &:= 52e_x + 36e_y & (1.16) \end{aligned}$$

Podemos calcular la **norma de un vector** o en otras palabras **su longitud** con el comando **Norm**. La definición de norma es

DEFINITION 6.2 *Norm of a Vector*

The norm, or magnitude, of a vector (a, b, c) is the number $\|(a, b, c)\|$ defined by

$$\|(a, b, c)\| = \sqrt{a^2 + b^2 + c^2}$$




FIGURE 6.2

$$\|(-1, 4, 1)\| = \sqrt{18}.$$

Ahora bien, calculemos la norma o longitud del siguiente vector

$$\begin{aligned} & \text{[> v18:= <-1,4,1>;} \\ & \qquad \qquad \qquad v18 := -e_x + 4e_y + e_z \end{aligned} \tag{1.17}$$

$$\begin{aligned} & \text{[> Norma_v18:= Norm(v18);} \\ & \qquad \qquad \qquad Norma_v18 := 3\sqrt{2} \end{aligned} \tag{1.18}$$

Calculemos tambien la norma de **v15a**

$$\begin{aligned} & \text{[> Norma_v15a:= Norm(v15a);} \\ & \qquad \qquad \qquad Norma_v15a := 20\sqrt{10} \end{aligned} \tag{1.19}$$

▼ Producto punto

Para calcular productos punto en coordenadas cartesianas MAPLE usa el comando **DotProduct** o el símbolo (**.**)

Considérese el producto de los vectores **v18** y de **v17a**

$$\begin{aligned} & \text{[> DotProduct(v18,v17a);} \\ & \qquad \qquad \qquad -65 \end{aligned} \tag{2.1}$$

y ahora con la otra opción

$$\begin{aligned} & \text{[> v18 . v17a;} \\ & \qquad \qquad \qquad -65 \end{aligned} \tag{2.2}$$

Pero hay que tomar en cuenta que el comando **solo funciona directamente para coordenadas cartesianas**. Para otros sistemas coordenados se hace como sigue

$$\begin{aligned} & \text{[> v19:= Vector([5,-5], coordinates=polar[r,theta]);} \\ & \qquad \qquad \qquad v19 := 5e_r - 5e_\theta \end{aligned} \tag{2.3}$$

$$\begin{aligned} & \text{[> DotProduct(v11,v19);} \\ & \text{Error, (in Student:-VectorCalculus:-DotProduct) cannot take} \\ & \text{the dot product of free Vectors in non-cartesian coordinates} \\ & \text{[> SetCoordinates(polar[r,theta]);} \\ & \qquad \qquad \qquad \text{polar}_{r,\theta} \end{aligned} \tag{2.4}$$

$$\begin{aligned} & \text{[> DotProduct(v11,v19);} \\ & \text{Error, (in Student:-VectorCalculus:-DotProduct) cannot take} \\ & \text{the dot product of free Vectors in non-cartesian coordinates} \\ & \text{[> v20 := RootedVector(root=[1,2],[a,b]);} \\ & \qquad \qquad \qquad v20 := \begin{bmatrix} a \\ b \end{bmatrix} \end{aligned} \tag{2.5}$$

$$\begin{aligned} & \text{[> v21 := RootedVector(root=[1,2],[c,d]);} \\ & \qquad \qquad \qquad \end{aligned} \tag{2.6}$$

$$v2I := \begin{bmatrix} c \\ d \end{bmatrix} \quad (2.6)$$

```
> DotProduct(v20,v21);
```

$$a c + b d \quad (2.7)$$

Para regresar al sistema cartesiano, se teclea lo siguiente

```
> SetCoordinates( cartesian[x,y,z] );
```

$$cartesian_{x,y,z} \quad (2.8)$$

Producto cruz

Para esta operación usamos el comando **CrossProduct** o en forma corta (**&x**)

La definición de producto cruz es

DEFINITION 6.7 Cross Product

The cross product of **F** with **G** is the vector **F** × **G** defined by

$$\mathbf{F} \times \mathbf{G} = (b_1 c_2 - b_2 c_1)\mathbf{i} + (a_2 c_1 - a_1 c_2)\mathbf{j} + (a_1 b_2 - a_2 b_1)\mathbf{k}.$$

De nueva cuenta, usemos los vectores **v18** y **v17a**

```
> v18;
v17a;
```

$$\begin{aligned} & -e_x + 4e_y + e_z \\ & 5e_x - 10e_y - 20e_z \end{aligned} \quad (3.1)$$

```
> CrossProduct(v18,v17a);
CrossProduct(v17a,v18);
```

$$\begin{aligned} & -70e_x - 15e_y - 10e_z \\ & 70e_x + 15e_y + 10e_z \end{aligned} \quad (3.2)$$

```
> v18 &x v17a;
v17a &x v18;
```

$$\begin{aligned} & -70e_x - 15e_y - 10e_z \\ & 70e_x + 15e_y + 10e_z \end{aligned} \quad (3.3)$$

y de nueva cuenta para coordenadas diferentes a las cartesianas el comando no funciona. Veamos,

```
> SetCoordinates( cylindrical );
```

$$cylindrical_{r,\theta,z} \quad (3.4)$$

```
> r1 := RootedVector([2,2,1], root=<1,2,3>);
```

$$(3.5)$$

$$r1 := \begin{bmatrix} 2 \\ 2 \\ 1 \end{bmatrix} \quad (3.5)$$

```
> r2 := RootedVector([3,-1,0], root=<1,2,3>);
```

$$r2 := \begin{bmatrix} 3 \\ -1 \\ 0 \end{bmatrix} \quad (3.6)$$

```
> r1 &x r2;
CrossProduct(r1,r2);
```

$$\begin{bmatrix} 1 \\ 3 \\ -8 \end{bmatrix}$$

$$\begin{bmatrix} 1 \\ 3 \\ -8 \end{bmatrix}$$

(3.7)

```
> SetCoordinates( cartesian[x,y,z] );
```

*cartesian*_{x,y,z}

(3.8)

Gradiente

Antes que nada, véase la definición de gradiente,

DEFINITION 12.8 Gradient

The gradient of a scalar field φ is the vector field $\nabla\varphi$ given by

$$\nabla\varphi = \frac{\partial\varphi}{\partial x}\mathbf{i} + \frac{\partial\varphi}{\partial y}\mathbf{j} + \frac{\partial\varphi}{\partial z}\mathbf{k},$$

wherever these partial derivatives are defined.

Es necesario decir que MAPLE provee 3 opciones para realizar este tipo de cálculos.

- **Del**: el cual define al operador nabla de la definición anterior ya sea en 2D o en 3D
- **Nabla**: al igual que antes, éste define al operador nabla de la definición anterior ya sea en 2D o en 3D
- **Gradient**: su definición es redundante

$$\text{Del=Nabla} = \frac{\partial}{\partial x} i + \frac{\partial}{\partial y} j + \frac{\partial}{\partial z} k$$

Consideremos entonces la siguiente función

$$\begin{aligned} &> \text{fxy10} := x^2 + 2y + xy; \\ &\text{fxy10} := x^2 + 2y + xy \end{aligned} \quad (4.1)$$

$$\begin{aligned} &> \text{grad_fxy10a} := \text{Del}(\text{fxy10}); \\ &\text{grad_fxy10b} := \text{Nabla}(\text{fxy10}); \\ &\text{grad_fxy10b} := \text{Gradient}(\text{fxy10}); \\ &\text{grad_fxy10a} := (2x + y)\bar{e}_x + (2 + x)\bar{e}_y \\ &\text{grad_fxy10b} := (2x + y)\bar{e}_x + (2 + x)\bar{e}_y \\ &\text{grad_fxy10b} := (2x + y)\bar{e}_x + (2 + x)\bar{e}_y \end{aligned} \quad (4.2)$$

Considérese ahora la siguiente función

$$\begin{aligned} &> \text{phi}[1] := x^2y - x\exp(z); \\ &\phi_1 := x^2y - e^z x \end{aligned} \quad (4.3)$$

$$\begin{aligned} &> \text{Del}(\text{phi}[1]); \\ &\text{Nabla}(\text{phi}[1]); \\ &\text{Gradient}(\text{phi}[1]); \\ &(2xy - e^z)\bar{e}_x + (x^2)\bar{e}_y - e^z x\bar{e}_z \\ &(2xy - e^z)\bar{e}_x + (x^2)\bar{e}_y - e^z x\bar{e}_z \\ &(2xy - e^z)\bar{e}_x + (x^2)\bar{e}_y - e^z x\bar{e}_z \end{aligned} \quad (4.4)$$

Divergencia

La divergencia de una función vectorial queda definida como

DEFINITION 12.10 Divergence

The divergence of a vector field $\mathbf{F}(x, y, z) = f(x, y, z)\mathbf{i} + g(x, y, z)\mathbf{j} + h(x, y, z)\mathbf{k}$ is the scalar field

$$\text{div } \mathbf{F} = \frac{\partial f}{\partial x} + \frac{\partial g}{\partial y} + \frac{\partial h}{\partial z}.$$

$$\text{Del} = \text{Nabla} = \frac{\partial}{\partial x} \mathbf{i} + \frac{\partial}{\partial y} \mathbf{j} + \frac{\partial}{\partial z} \mathbf{k}$$

La divergencia se puede calcular a partir de las siguientes opciones

- **Divergence**: es la forma natural de nombrar este comando
- **Del.**: que resulta obvio del producto punto del operador definido anteriormente con una función vectorial
- **Nabla.**: que resulta obvio del producto punto del operador definido anteriormente con una función vectorial
- A pie: es decir, si se es capaz de escribir un vector, derivadas y se conocen las componentes de la

función, en realidad los comandos anteriores salen sobrando.

Consideremos la siguiente función vectorial $\mathbf{F} = x\mathbf{i} + y\mathbf{j} + 2z\mathbf{k}$

$$\begin{aligned} & \text{[> F00:= VectorField(<x,y,2*z>);} \\ & \qquad \qquad \qquad F00 := (x)\bar{e}_x + (y)\bar{e}_y + 2z\bar{e}_z \end{aligned} \tag{5.1}$$

$$\begin{aligned} & \text{[> Divergence(F00);} \\ & \qquad \qquad \qquad 4 \end{aligned} \tag{5.2}$$

$$\begin{aligned} & \text{[> Del. F00;} \\ & \qquad \qquad \qquad 4 \end{aligned} \tag{5.3}$$

$$\begin{aligned} & \text{[> Nabla. F00;} \\ & \qquad \qquad \qquad 4 \end{aligned} \tag{5.4}$$

RETO A PROFESORES

Se les propone a los profesores, que en base a la información sobre:

- escritura de una función vectorial
- escritura de una derivada
- las componentes de la función **F00**

escriban y realicen en MAPLE la operación $\nabla \cdot \mathbf{F00}$ SIN AYUDA DE LOS COMANDOS: Divergence, Del. Nabla.

Consideremos ahora la función **F10** vectorial definida al inicio

$$\begin{aligned} & \text{[> Divergence(F10);} \\ & \qquad \qquad \qquad 3x \end{aligned} \tag{5.5}$$

$$\begin{aligned} & \text{[> Del. F10;} \\ & \qquad \qquad \qquad 3x \end{aligned} \tag{5.6}$$

$$\begin{aligned} & \text{[> Nabla. F10;} \\ & \qquad \qquad \qquad 3x \end{aligned} \tag{5.7}$$

Consideremos ahora la divergencia de un gradiente. Tomemos por ejemplo la función **phi[1]**

$$\begin{aligned} & \text{[> phi[1];} \\ & \qquad \qquad \qquad x^2y - e^z x \end{aligned} \tag{5.8}$$

$$\begin{aligned} & \text{[> Divergence(Gradient(phi[1]));} \\ & \qquad \qquad \qquad 2y - e^z x \end{aligned} \tag{5.9}$$

$$\begin{aligned} & \text{[> Del.Del(phi[1]);} \\ & \qquad \qquad \qquad 2y - e^z x \end{aligned} \tag{5.10}$$

Rotacional (Curl)

La definición del rotacional es como sigue,

DEFINITION 12.11 Curl

The curl of a vector field $\mathbf{F}(x, y, z) = f(x, y, z)\mathbf{i} + g(x, y, z)\mathbf{j} + h(x, y, z)\mathbf{k}$ is the vector field

$$\text{curl } \mathbf{F} = \left(\frac{\partial h}{\partial y} - \frac{\partial g}{\partial z} \right) \mathbf{i} + \left(\frac{\partial f}{\partial z} - \frac{\partial h}{\partial x} \right) \mathbf{j} + \left(\frac{\partial g}{\partial x} - \frac{\partial f}{\partial y} \right) \mathbf{k}.$$

$$\text{Del} = \text{Nabla} = \frac{\partial}{\partial x} \mathbf{i} + \frac{\partial}{\partial y} \mathbf{j} + \frac{\partial}{\partial z} \mathbf{k}$$

Sabemos que el rotacional resulta de un producto cruz entre el operador anterior y una función vectorial. En MAPLE podemos usar los siguientes comandos,

- **Curl**: es la forma natural de nombrar al comando
- **Del &x**: resulta de la definición de producto cruz
- **Nabla &x**: resulta de la definición de producto cruz
- A pie: s decir, si se es capaz de escribir un vector, derivadas y se conocen las componentes de la función, en realidad los comandos anteriores salen sobrando.

Consideremos entonces, la siguiente función $\mathbf{F} = y\mathbf{i} + 2xz\mathbf{j} + ze^x\mathbf{k}$

$$\begin{aligned} > \text{F01} := \text{VectorField}(<y, 2*x*z, z*\exp(x)>); \\ & \quad F01 := (y)\bar{e}_x + 2xz\bar{e}_y + (ze^x)\bar{e}_z \end{aligned} \quad (6.1)$$

$$\begin{aligned} > \text{Curl}(\text{F01}); \\ & \quad -2x\bar{e}_x - ze^x\bar{e}_y + (2z-1)\bar{e}_z \end{aligned} \quad (6.2)$$

$$\begin{aligned} > \text{Del } \&x \text{ F01}; \\ & \quad -2x\bar{e}_x - ze^x\bar{e}_y + (2z-1)\bar{e}_z \end{aligned} \quad (6.3)$$

$$\begin{aligned} > \text{Nabla } \&x \text{ F01}; \\ & \quad -2x\bar{e}_x - ze^x\bar{e}_y + (2z-1)\bar{e}_z \end{aligned} \quad (6.4)$$

RETO A PROFESORES

Se les propone a los profesores, que en base a la información sobre:

- escritura de una función vectorial
- escritura de una derivada
- las componentes de la función **F00**

escriban y realicen en MAPLE la operación $\nabla \times \mathbf{F01}$ SIN AYUDA DE LOS COMANDOS: Divergence, Del, Nabla.

Consideremos ahora la siguiente función vectorial

$$> \text{F02} := \text{VectorField}(<4*\sin(x*y), \cos(x^2*y), 0>);$$

$$F02 := 4 \sin(xy) \bar{e}_x + (\cos(x^2 y)) \bar{e}_y \quad (6.5)$$

> Curl(F02);

$$(-2 \sin(x^2 y) xy - 4 \cos(xy) x) \bar{e}_z \quad (6.6)$$

Verifiquemos ahora la siguiente propiedad vectorial $\nabla \times (\nabla \phi) = 0$. Usaremos la función **phi[1]**

> phi[1];

$$x^2 y - e^z x \quad (6.7)$$

> Curl(Gradient(phi[1]));

$$0 \bar{e}_x \quad (6.8)$$

> Del &x Del(phi[1]);

$$0 \bar{e}_x \quad (6.9)$$

Derivadas de funciones vectoriales

La función de derivada está siempre funcionando en MAPLE; pero cuando se trata de funciones es otra cosa. No se pueden derivar funciones vectoriales amén de que esté cargado el paquete de **VectorCalculus**

En realidad la explicación de este comando es bastante simple ya que es **diff**

Considérese la función vectorial **F01**

> F01;

$$(y) \bar{e}_x + 2 x z \bar{e}_y + (z e^x) \bar{e}_z \quad (7.1)$$

> diff(F01, x);

$$2 z \bar{e}_y + (z e^x) \bar{e}_z \quad (7.2)$$

> diff(F01, y);

$$\bar{e}_x \quad (7.3)$$

> diff(F01, x, y);

$$0 \bar{e}_x \quad (7.4)$$

Por otro lado, el comando diff permite derivar funciones definidas como simples vectores y aquellas definidas con **VectorField**

> <t, t^2, t^3>;

$$(t) e_x + (t^2) e_y + (t^3) e_z \quad (7.5)$$

> diff(<t, t^2, t^3> , t);

$$e_x + 2 t e_y + 3 t^2 e_z \quad (7.6)$$

Pero téngase en cuenta que para cálculos con **Divergence**, **Curl** y **Gradient** las funciones deben ser estrictamente definidas con **VectorField**

Véase otro ejemplo,

$$\begin{aligned} & \text{> diff(<cos(s*t), sin(s*t)>, s, t);} \\ & \quad (-\cos(st) st - \sin(st))e_x + (-\sin(st) st + \cos(st))e_y \end{aligned} \quad (7.7)$$

Integrales de línea

La definición de integral de línea es como sigue,

DEFINITION 13.1 Line Integral

Suppose a smooth curve C has coordinate functions $x = x(t)$, $y = y(t)$, $z = z(t)$ for $a \leq t \leq b$. Let $f(x, y, z)$, $g(x, y, z)$, and $h(x, y, z)$ be continuous at least on the graph of C . Then, the line integral

$$\int_C f(x, y, z) dx + g(x, y, z) dy + h(x, y, z) dz$$

is defined by

$$\begin{aligned} & \int_C f(x, y, z) dx + g(x, y, z) dy + h(x, y, z) dz \\ &= \int_a^b \left[f(x(t), y(t), z(t)) \frac{dx}{dt} + g(x(t), y(t), z(t)) \frac{dy}{dt} + h(x(t), y(t), z(t)) \frac{dz}{dt} \right] dt. \end{aligned}$$

Consideremos un ejemplo en el que se considera una la integral de línea de una función vectorial a lo largo de una línea.

$$\begin{aligned} & \text{> F03:= VectorField(<x,y>);} \\ & \quad F03 := (x)\bar{e}_x + (y)\bar{e}_y \end{aligned} \quad (8.1)$$

La curva en este caso es una línea que va del punto (1,2) al punto (3,-4). En MAPLE la definimos como

$$\begin{aligned} & \text{> Clinea00:= Line(<1,2>,<3,-4>);} \\ & \quad Clinea00 := \text{Line}(e_x + 2e_y, 3e_x - 4e_y) \end{aligned} \quad (8.2)$$

Para calcular la integral de línea usamos el comando **LineInt**


$$\begin{aligned} & \text{> LineInt(F03,Clinea00);} \\ & \quad 10 \end{aligned} \quad (8.3)$$

Podemos visualizar la integral también

$$\begin{aligned} & \text{> LineInt(F03,Clinea00, output=integral);} \\ & \quad \int_0^1 (-10 + 40t) dt \end{aligned} \quad (8.4)$$

Podemos visualizar el gráfico

```
> LineInt(F03,Clinea00, output=plot);
```



The path of integration, vector(s) tangent to the path, and vector-field arrows

Considérese ahora el siguiente ejemplo,

EXAMPLE 13.6

Calculate the work done by $\mathbf{F} = \mathbf{i} - y\mathbf{j} + xyz\mathbf{k}$ in moving a particle from $(0, 0, 0)$ to $(1, -1, 1)$ along the curve $x = t$, $y = -t^2$, $z = t$ for $0 \leq t \leq 1$.

Comencemos por definir el campo vectorial $\mathbf{F}$

```
> f04:= 1;
   g04:= -y;
   h04:= x*y*x;
```

$$\begin{aligned} f04 &:= 1 \\ g04 &:= -y \\ h04 &:= x^2 y \end{aligned}$$

(8.5)

```
> F04:= VectorField(<f04,g04,h04>);
```

$$F04 := \bar{e}_x - y\bar{e}_y + (x^2 y)\bar{e}_z$$

(8.6)

La curva la podemos definir como:

```
> x04:= t;  
y04:= -t^2;  
z04:= t;
```

$$\begin{aligned}x04 &:= t \\ y04 &:= -t^2 \\ z04 &:= t\end{aligned}$$

(8.7)


```
> R04:= PositionVector([x04,y04,z04]);
```

$$R04 := \begin{bmatrix} t \\ -t^2 \\ t \end{bmatrix}$$

(8.8)

y lo graficamos como sigue

```
> PlotPositionVector(R04,t=0..1);
```



La integral de línea se calcula como sigue


```
> LineInt(F04, Path(R04, t = 0 .. 1));
```

$$\frac{3}{10}$$

(8.9)

Para visualizar la integral tenemos


```
> LineInt(F04, Path(R04, t = 0 .. 1), output=integral);
```

$$\int_0^1 (1 - 2t^3 - t^4) dt$$

(8.10)

Para visualizar el gráfico tenemos:

```
> LineInt(F04, Path(R04, t = 0 .. 1), output=plot, axes=boxed);
```



The path of integration, vector(s) tangent to the path, and vector-field arrows

```
>
```

Herramientas de graficación


Se muestran algunos comandos útiles para graficar.

Por ejemplo para ver la intersección entre un plano y una superficie se usa el comando

CrossSection


```
> with(Student[MultivariateCalculus]):  
> CrossSection(x^2+y^2+z^2 = 4, x+y= [1.1,0.1], x=-2..2, y=-2..2,  
  z=-2..2, showfunction = false, title="Sphere");
```

Sphere



Level curves of the surface $f(x, y, z) = x^2 + y^2 + z^2 - 4$ and one or more planes of the form $x + y = \text{constant}$.


```
> CrossSection(x+y^2, x+y-z=0, x=2..4, y=-1..2);
```




The intersection of the surface $f(x, y) = x + y^2$ and one or more planes of the form $x + y - z = \text{constant}$.

Para graficar curvas de contorno, se usa **contourplot**

```
> with(plots):  
> contourplot(sin(x*y), x=-3..3, y=-3..3);
```




```
> contourplot(-5*x/(x^2 + y^2 + 1), x=-3..3, y=-3..3,  
filledregions=true,  
coloring=["White", "DarkViolet"]);
```



>

▼ Tutor de campos vectoriales

```
> with(Student[VectorCalculus]):  
> VectorFieldTutor();
```



>

▼ Tutor de curvas

```
> with(Student[VectorCalculus]):
> SpaceCurveTutor();
```

